

NAG Library Routine Document

G07CAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of **bold italicised** terms and other implementation-dependent details.

1 Purpose

G07CAF computes a t -test statistic to test for a difference in means between two Normal populations, together with a confidence interval for the difference between the means.

2 Specification

```

SUBROUTINE G07CAF(TAIL, EQUAL, NX, NY, XMEAN, YMEAN, XSTD, YSTD, CLEVEL,
1              T, DF, PROB, DL, DU, IFAIL)
      INTEGER          NX, NY, IFAIL
      double precision XMEAN, YMEAN, XSTD, YSTD, CLEVEL, T, DF, PROB, DL, DU
      CHARACTER*1      TAIL, EQUAL

```

3 Description

Consider two independent samples, denoted by X and Y , of size n_x and n_y drawn from two Normal populations with means μ_x and μ_y , and variances σ_x^2 and σ_y^2 respectively. Denote the sample means by $\bar{x}$ and $\bar{y}$ and the sample variances by s_x^2 and s_y^2 respectively.

G07CAF calculates a test statistic and its significance level to test the null hypothesis $H_0 : \mu_x = \mu_y$, together with upper and lower confidence limits for $\mu_x - \mu_y$. The test used depends on whether or not the two population variances are assumed to be equal.

1. It is assumed that the two variances are equal, that is $\sigma_x^2 = \sigma_y^2$.

The test used is the two sample t -test. The test statistic t is defined by;

$$t_{\text{obs}} = \frac{\bar{x} - \bar{y}}{s \sqrt{(1/n_x) + (1/n_y)}}$$

where

$$s^2 = \frac{(n_x - 1)s_x^2 + (n_y - 1)s_y^2}{n_x + n_y - 2}$$

is the pooled variance of the two samples.

Under the null hypothesis H_0 this test statistic has a t -distribution with $(n_x + n_y - 2)$ degrees of freedom.

The test of H_0 is carried out against one of three possible alternatives;

$H_1 : \mu_x \neq \mu_y$; the significance level, $p = P(t \geq |t_{\text{obs}}|)$, i.e., a two tailed probability.

$H_1 : \mu_x > \mu_y$; the significance level, $p = P(t \geq t_{\text{obs}})$, i.e., an upper tail probability.

$H_1 : \mu_x < \mu_y$; the significance level, $p = P(t \leq t_{\text{obs}})$, i.e., a lower tail probability.

Upper and lower $100(1 - \alpha)\%$ confidence limits for $\mu_x - \mu_y$ are calculated as:

$$(\bar{x} - \bar{y}) \pm t_{1-\alpha/2} s \sqrt{(1/n_x) + (1/n_y)}.$$

where $t_{1-\alpha/2}$ is the $100(1 - \alpha/2)$ percentage point of the t -distribution with $(n_x + n_y - 2)$ degrees of freedom.

2. It is not assumed that the two variances are equal.

If the population variances are not equal the usual two sample t -statistic no longer has a t -distribution and an approximate test is used.

This problem is often referred to as the Behrens–Fisher problem, see Kendall and Stuart (1969). The test used here is based on Satterthwaites procedure. To test the null hypothesis the test statistic t' is used where

$$t'_{\text{obs}} = \frac{\bar{x} - \bar{y}}{\text{se}(\bar{x} - \bar{y})}$$

$$\text{where } \text{se}(\bar{x} - \bar{y}) = \sqrt{\frac{s_x^2}{n_x} + \frac{s_y^2}{n_y}}.$$

A t -distribution with f degrees of freedom is used to approximate the distribution of t' where

$$f = \frac{\text{se}(\bar{x} - \bar{y})^4}{\frac{(s_x^2/n_x)^2}{(n_x - 1)} + \frac{(s_y^2/n_y)^2}{(n_y - 1)}}.$$

The test of H_0 is carried out against one of the three alternative hypotheses described above, replacing t by t' and t_{obs} by t'_{obs} .

Upper and lower $100(1 - \alpha)\%$ confidence limits for $\mu_x - \mu_y$ are calculated as:

$$(\bar{x} - \bar{y}) \pm t_{1-\alpha/2} \text{se}(\bar{x} - \bar{y}).$$

where $t_{1-\alpha/2}$ is the $100(1 - \alpha/2)$ percentage point of the t -distribution with f degrees of freedom.

4 References

Johnson M G and Kotz A (1969) *The Encyclopedia of Statistics* 2 Griffin

Kendall M G and Stuart A (1969) *The Advanced Theory of Statistics (Volume 1)* (3rd Edition) Griffin

Snedecor G W and Cochran W G (1967) *Statistical Methods* Iowa State University Press

5 Parameters

- 1: TAIL – CHARACTER*1 Input

On entry: indicates which tail probability is to be calculated, and thus which alternative hypothesis is to be used.

TAIL = 'T'

The two tail probability, i.e., $H_1 : \mu_x \neq \mu_y$.

TAIL = 'U'

The upper tail probability, i.e., $H_1 : \mu_x > \mu_y$.

TAIL = 'L'

The lower tail probability, i.e., $H_1 : \mu_x < \mu_y$.

Constraint: TAIL = 'T', 'U' or 'L'.

- 2: EQUAL – CHARACTER*1 Input

On entry: indicates whether the population variances are assumed to be equal or not.

EQUAL = 'E'

The population variances are assumed to be equal, that is $\sigma_x^2 = \sigma_y^2$.

EQUAL = 'U'

The population variances are not assumed to be equal.

Constraint: EQUAL = 'E' or 'U'.

- | | | |
|-----|--|---------------------|
| 3: | NX – INTEGER | <i>Input</i> |
| | <i>On entry:</i> n_x , the size of the X sample. | |
| | <i>Constraint:</i> $NX \geq 2$. | |
| 4: | NY – INTEGER | <i>Input</i> |
| | <i>On entry:</i> n_y , the size of the Y sample. | |
| | <i>Constraint:</i> $NY \geq 2$. | |
| 5: | XMEAN – double precision | <i>Input</i> |
| | <i>On entry:</i> $\bar{x}$, the mean of the X sample. | |
| 6: | YMEAN – double precision | <i>Input</i> |
| | <i>On entry:</i> $\bar{y}$, the mean of the Y sample. | |
| 7: | XSTD – double precision | <i>Input</i> |
| | <i>On entry:</i> s_x , the standard deviation of the X sample. | |
| | <i>Constraint:</i> $XSTD > 0.0$. | |
| 8: | YSTD – double precision | <i>Input</i> |
| | <i>On entry:</i> s_y , the standard deviation of the Y sample. | |
| | <i>Constraint:</i> $YSTD > 0.0$. | |
| 9: | CLEVEL – double precision | <i>Input</i> |
| | <i>On entry:</i> the confidence level, $1 - \alpha$, for the specified tail. For example CLEVEL = 0.95 will give a 95% confidence interval. | |
| | <i>Constraint:</i> $0.0 < \text{CLEVEL} < 1.0$. | |
| 10: | T – double precision | <i>Output</i> |
| | <i>On exit:</i> contains the test statistic, t_{obs} or t'_{obs} . | |
| 11: | DF – double precision | <i>Output</i> |
| | <i>On exit:</i> contains the degrees of freedom for the test statistic. | |
| 12: | PROB – double precision | <i>Output</i> |
| | <i>On exit:</i> contains the significance level, that is the tail probability, p , as defined by TAIL. | |
| 13: | DL – double precision | <i>Output</i> |
| | <i>On exit:</i> contains the lower confidence limit for $\mu_x - \mu_y$. | |
| 14: | DU – double precision | <i>Output</i> |
| | <i>On exit:</i> contains the upper confidence limit for $\mu_x - \mu_y$. | |
| 15: | IFAIL – INTEGER | <i>Input/Output</i> |
| | <i>On entry:</i> IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details. | |

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0 . **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, TAIL $\neq$ 'T', 'U' or 'L',
 or EQUAL $\neq$ 'E' or 'U',
 or NX < 2 ,
 or NY < 2 ,
 or XSTD ≤ 0.0 ,
 or YSTD ≤ 0.0 ,
 or CLEVEL ≤ 0.0 ,
 or CLEVEL ≥ 1.0 .

7 Accuracy

The computed probability and the confidence limits should be accurate to approximately five significant figures.

8 Further Comments

The sample means and standard deviations can be computed using G01AAF.

9 Example

This example reads the two sample sizes and the sample means and standard deviations for two independent samples. The data is taken from page 116 of Snedecor and Cochran (1967) from a test to compare two methods of estimating the concentration of a chemical in a vat. A test of the equality of the means is carried out first assuming that the two population variances are equal and then making no assumption about the equality of the population variances.

9.1 Program Text

```
*      G07CAF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*      .. Local Scalars ..
      DOUBLE PRECISION CLEVEL, DF, DL, DU, PROB, T, XMEAN, XSTD, YMEAN,
+      YSTD
      INTEGER          IFAIL, NX, NY
*      .. External Subroutines ..
      EXTERNAL         G07CAF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G07CAF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) NX, NY
      READ (NIN,*) XMEAN, YMEAN, XSTD, YSTD
```

```

      READ (NIN,*) CLEVEL
      IFAIL = 0
*
      CALL G07CAF('Two','Equal',NX,NY,XMEAN,YMEAN,XSTD,YSTD,CLEVEL,T,DF,
+               PROB,DL,DU,IFAIL)
*
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Assuming population variances are equal.'
      WRITE (NOUT,*)
      WRITE (NOUT,99999) 't test statistic = ', T
      WRITE (NOUT,99998) 'Degrees of freedom = ', DF
      WRITE (NOUT,99997) 'Significance level = ', PROB
      WRITE (NOUT,99999)
+ 'Lower confidence limit for difference in means = ', DL
      WRITE (NOUT,99999)
+ 'Upper confidence limit for difference in means = ', DU
      WRITE (NOUT,*)
      IFAIL = 0
*
      CALL G07CAF('Two','Unequal',NX,NY,XMEAN,YMEAN,XSTD,YSTD,CLEVEL,T,
+               DF,PROB,DL,DU,IFAIL)
*
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'No assumptions about population variances .'
      WRITE (NOUT,*)
      WRITE (NOUT,99999) 't test statistic = ', T
      WRITE (NOUT,99997) 'Degrees of freedom = ', DF
      WRITE (NOUT,99997) 'Significance level = ', PROB
      WRITE (NOUT,99999)
+ 'Lower confidence limit for difference in means = ', DL
      WRITE (NOUT,99999)
+ 'Upper confidence limit for difference in means = ', DU
*
99999 FORMAT (1X,A,F10.4)
99998 FORMAT (1X,A,F8.1)
99997 FORMAT (1X,A,F8.4)
      END

```

9.2 Program Data

G07CAF Example Program Data

```

4  8
25.0 21.0
0.8185 4.2083
0.95

```

9.3 Program Results

G07CAF Example Program Results

Assuming population variances are equal.

```

t test statistic =      1.8403
Degrees of freedom =      10.0
Significance level =      0.0955
Lower confidence limit for difference in means =      -0.8429
Upper confidence limit for difference in means =      8.8429

```

No assumptions about population variances .

```

t test statistic =      2.5922
Degrees of freedom =      7.9925
Significance level =      0.0320
Lower confidence limit for difference in means =      0.4410
Upper confidence limit for difference in means =      7.5590

```